
Subject: making a checkerboard array?

Posted by [Mike\[2\]](#) on Tue, 25 Sep 2007 16:27:17 GMT

[View Forum Message](#) <> [Reply to Message](#)

I'm trying to make a checkerboard mask for an array, but I'm missing something that is likely to be obvious to the IDL array masters. The following code makes a pattern like this (use a fixed width font):

```
+++++
+++++
+++++
00000+++++00000+++++00000+++++
00000+++++00000+++++00000+++++
00000+++++00000+++++00000+++++
+++++
+++++
+++++
00000+++++00000+++++00000+++++
00000+++++00000+++++00000+++++
00000+++++00000+++++00000+++++
+++++
+++++
+++++
00000+++++00000+++++00000+++++
00000+++++00000+++++00000+++++
00000+++++00000+++++00000+++++
```

Can anyone help me fill in the missing rectangles like this?

```
+++++00000+++++00000+++++00000
+++++00000+++++00000+++++00000
+++++00000+++++00000+++++00000
00000+++++00000+++++00000+++++
00000+++++00000+++++00000+++++
00000+++++00000+++++00000+++++
+++++00000+++++00000+++++00000
+++++00000+++++00000+++++00000
+++++00000+++++00000+++++00000
00000+++++00000+++++00000+++++
00000+++++00000+++++00000+++++
00000+++++00000+++++00000+++++
+++++00000+++++00000+++++00000
+++++00000+++++00000+++++00000
+++++00000+++++00000+++++00000
00000+++++00000+++++00000+++++
00000+++++00000+++++00000+++++
00000+++++00000+++++00000+++++
```

```
Nx=300
Ny=200
Nrects = 10
Xside = Nx/Nrects
Yside = Ny/Nrects
xs = [replicate(1,Xside), replicate(0,Xside)]
while n_elements(xs) lt Nx do xs = [xs, [replicate(1,Xside),
replicate(0,Xside)]]
xs = xs[0:Nx-1]
ys = [replicate(1,Yside), replicate(0,Yside)]
while n_elements(ys) lt Ny do ys = [ys, [replicate(1,Yside),
replicate(0,Yside)]]
ys = ys[0:Ny-1]
window, xsize=Nx, ysize=Ny
tvsc1, xs # ys
```

Subject: Re: making a checkerboard array?
Posted by [David Fanning](#) on Tue, 25 Sep 2007 16:56:22 GMT
[View Forum Message](#) <> [Reply to Message](#)

Mike writes:

> I'm trying to make a checkerboard mask for an array, but I'm missing
> something that is likely to be obvious to the IDL array masters. The
> following code makes a pattern like this (use a fixed width font):

Here is my implementation:

```
Function Checkerboard, xsize, ysize, white, black
```

```
IF N_Elements(xsize) EQ 0 THEN xsize = 512
IF N_Elements(ysize) EQ 0 THEN ysize = 512
IF N_Elements(white) EQ 0 THEN white = 255
IF N_Elements(black) EQ 0 THEN black = 0
```

```
IF xsize MOD 2 NE 0 THEN $
  Message, 'The X size must be an even number.'
IF ysize MOD 2 NE 0 THEN $
  Message, 'The Y size must be an even number.'
```

```
horizontalBoard = FltArr(xsize, ysize)
verticalboard = FltArr(xsize, ysize)
xhalfSize = xsize / 2
x = IndGen(xhalfSize) * 2
yhalfSize = ysize / 2
y = IndGen(yhalfsize) * 2
```

```

horizontalBoard[x,*] = 1
verticalBoard[:,y] = 1
verticalBoard = Temporary(verticalBoard) + horizontalBoard
verticalBoard[Where(verticalBoard EQ 2)] = 0
board = FltArr(xsize, ysize) + white
board[Where(verticalBoard GT 0)] = black
RETURN, board
END

```

Are you working through Gonzales and Woods? My 3rd Edition just arrived yesterday! :-)

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.dfanning.com/>

Sepore ma de ni thui. ("Perhaps thou speakest truth.")

Subject: Re: making a checkerboard array?

Posted by [JD Smith](#) on Tue, 25 Sep 2007 21:34:54 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Tue, 25 Sep 2007 09:27:17 -0700, Mike wrote:

> I'm trying to make a checkerboard mask for an array, but I'm missing
 > something that is likely to be obvious to the IDL array masters.

> Can anyone help me fill in the missing rectangles like this?

>

```

> +++++00000+++++00000+++++00000
> +++++00000+++++00000+++++00000
> +++++00000+++++00000+++++00000
> 00000+++++00000+++++00000+++++
> 00000+++++00000+++++00000+++++
> 00000+++++00000+++++00000+++++
> +++++00000+++++00000+++++00000
> +++++00000+++++00000+++++00000
> +++++00000+++++00000+++++00000
> 00000+++++00000+++++00000+++++
> 00000+++++00000+++++00000+++++
> 00000+++++00000+++++00000+++++
> +++++00000+++++00000+++++00000
> +++++00000+++++00000+++++00000

```

```
> +++++00000+++++00000+++++00000
> 00000+++++00000+++++00000+++++
> 00000+++++00000+++++00000+++++
> 00000+++++00000+++++00000+++++
```

```
l=lindgen(nx,ny)
```

```
l=(l mod (xside*2) lt xside) XOR (l/nx mod (yside*2) ge yside)
```

JD

Subject: Re: making a checkerboard array?

Posted by [David Fanning](#) on Tue, 25 Sep 2007 21:52:33 GMT

[View Forum Message](#) <> [Reply to Message](#)

JD Smith writes:

```
>> I'm trying to make a checkerboard mask for an array, but I'm missing
>> something that is likely to be obvious to the IDL array masters.
```

```
>
```

```
> l=lindgen(nx,ny)
```

```
> l=(l mod (xside*2) lt xside) XOR (l/nx mod (yside*2) ge yside)
```

Well, uh, it's not obvious to me. :-(

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.dfanning.com/>

Sepore ma de ni thui. ("Perhaps thou speakest truth.")

Subject: Re: making a checkerboard array?

Posted by [Jean H.](#) on Tue, 25 Sep 2007 22:07:41 GMT

[View Forum Message](#) <> [Reply to Message](#)

David Fanning wrote:

```
> JD Smith writes:
```

```
>
```

```
>
```

```
>>> I'm trying to make a checkerboard mask for an array, but I'm missing
```

```
>>> something that is likely to be obvious to the IDL array masters.
```

```
>>
```

```
>> l=lindgen(nx,ny)
```

```
>> l=(l mod (xside*2) lt xside) XOR (l/nx mod (yside*2) ge yside)
```

```
>
```

```
>
```

```
> Well, uh, it's not obvious to me. :-(
```

```
>
```

```
> Cheers,
```

```
>
```

```
> David
```

```
IDL> nx=20
```

```
IDL> ny=20
```

```
IDL> l=lindgen(nx,ny)
```

```
IDL> l=(l mod (nx*2) lt nx) XOR (l/nx mod (ny*2) ge ny)
```

```
IDL> print,l
```

```
  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1
1  1  1
  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0
0  0  0
  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1
1  1  1
  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0
0  0  0
  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1
1  1  1
  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0
0  0  0
  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1
1  1  1
  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0
0  0  0
  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1
1  1  1
  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0
0  0  0
  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1
1  1  1
  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0
0  0  0
  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1
1  1  1
  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0
0  0  0
  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1  1
1  1  1
  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0
```

```

0 0 0
  1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
1 1 1
  0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0

```

Subject: Re: making a checkerboard array?

Posted by [David Fanning](#) on Tue, 25 Sep 2007 22:59:57 GMT

[View Forum Message](#) <> [Reply to Message](#)

Jean H writes:

```

> IDL> nx=20
> IDL> ny=20
> IDL> l=lindgen(nx,ny)
> IDL> l=(l mod (nx*2) lt nx) XOR (l/nx mod (ny*2) ge ny)
> IDL> print,l

```

If there is a question in there, I think this is the answer:

```

IDL> nx=20
IDL> ny=20
IDL> l=lindgen(nx, ny)
IDL> l=(l mod (2) lt 1) XOR (l/nx mod (2) ge 1)
IDL> print, l
  1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0
1 0
  0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1
0 1
  1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0
1 0
  0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1
0 1
  1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0
1 0
  0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1
0 1
  1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0
1 0
  0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1
0 1
  1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0
1 0
  0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1
0 1
  1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0
1 0

```

```

0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1
0 1
1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0
1 0
0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1
0 1
1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0
1 0
0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1
0 1
1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0
1 0
0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1
0 1
1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0
1 0
0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1 0 1
0 1

```

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.dfanning.com/>

Sepore ma de ni thui. ("Perhaps thou speakest truth.")

Subject: Re: making a checkerboard array?

Posted by [Mike\[2\]](#) on Wed, 26 Sep 2007 14:46:07 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Sep 25, 5:34 pm, JD Smith <jdsm...@as.arizona.edu> wrote:

>

> l=lindgen(nx,ny)

> l=(l mod (xside*2) lt xside) XOR (l/nx mod (yside*2) ge yside)

I'm speechless - thanks!

Mike

Subject: Re: making a checkerboard array?

Posted by [Mike\[2\]](#) on Wed, 26 Sep 2007 15:02:07 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Sep 25, 12:56 pm, David Fanning <n...@dfanning.com> wrote:

> Are you working through Gonzales and Woods? My 3rd Edition just
> arrived yesterday! :-)

No - I'm making an image fuser that might work better than alpha
blending for pairs of PET images. Your solution almost works. If I
congrid it to make the squares bigger, it is just right. But JD's
is ... well ... I have no idea how a mind just pops that out!

Mike

P.S. Someone who is happy to get his new image processing book may
appreciate the quiz that is going around the lab here:
http://www.nerdtests.com/ft_nq.php ;-)

Subject: Re: making a checkerboard array?
Posted by [Bruce Bowler](#) on Wed, 26 Sep 2007 15:19:36 GMT
[View Forum Message](#) <> [Reply to Message](#)

> http://www.nerdtests.com/ft_nq.php ;-)

Arghhhhhh 91. I didn't think I was *that* nerdy :-)

--

+-----+-----+
Bruce Bowler | Those who cast the votes decide nothing. Those who
1.207.633.9600 | count the votes decide everything. - Anon
bbowler@bigelow.org |
+-----+-----+

Subject: Re: making a checkerboard array?
Posted by [David Fanning](#) on Wed, 26 Sep 2007 15:28:10 GMT
[View Forum Message](#) <> [Reply to Message](#)

Bruce Bowler writes:

> Arghhhhhh 91. I didn't think I was *that* nerdy :-)

Are you available for some map documentation translation work?

Cheers,

David

--

David Fanning, Ph.D.
Fanning Software Consulting, Inc.
Coyote's Guide to IDL Programming: <http://www.dfanning.com/>
Sepore ma de ni thui. ("Perhaps thou speakest truth.")

Subject: Re: making a checkerboard array?
Posted by [David Fanning](#) on Wed, 26 Sep 2007 15:29:25 GMT
[View Forum Message](#) <> [Reply to Message](#)

Bruce Bowler writes:

> Arghhhhhh 91. I didn't think I was *that* nerdy :-)

Let's just say in college I was an English major and
let it go at that. :-(

Cheers,

David

--

David Fanning, Ph.D.
Fanning Software Consulting, Inc.
Coyote's Guide to IDL Programming: <http://www.dfanning.com/>
Sepore ma de ni thui. ("Perhaps thou speakest truth.")

Subject: Re: making a checkerboard array?
Posted by [Allan Whiteford](#) on Wed, 26 Sep 2007 16:37:24 GMT
[View Forum Message](#) <> [Reply to Message](#)

Bruce Bowler wrote:

>> http://www.nerdtests.com/ft_nq.php ;-)

>

>

> Arghhhhhh 91. I didn't think I was *that* nerdy :-)

>

The test is scoring people too highly... I don't deserve the 94 I got.
Maybe a few years ago, but not now... surely not now!?!

Maybe it's just English majors bringing the average down ;).

I could tell I was heading for a high score when there wasn't an option
to pick "None of the above, it should be a lower case 'c'" which is what
I really wanted to answer.

Thanks,

Allan

Subject: Re: making a checkerboard array?

Posted by [R.G.Stockwell](#) on Wed, 26 Sep 2007 16:44:17 GMT

[View Forum Message](#) <> [Reply to Message](#)

"Bruce Bowler" <bbowler@bigelow.org> wrote in message
news:pan.2007.09.26.15.19.33@bigelow.org...

>> http://www.nerdtests.com/ft_nq.php ;-)

>

> Arghhhhhh 91. I didn't think I was **that** nerdy :-)

OHOH!

{takes test results, prints them to printer, and burns them! Pleasantly
dreams of
scoring only 91}

Subject: Re: making a checkerboard array?

Posted by [JD Smith](#) on Wed, 26 Sep 2007 17:27:37 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Tue, 25 Sep 2007 15:52:33 -0600, David Fanning wrote:

> JD Smith writes:

>

>>> I'm trying to make a checkerboard mask for an array, but I'm missing

>>> something that is likely to be obvious to the IDL array masters.

>>

>> `l=lindgen(nx,ny)`

>> `l=(l mod (xside*2) lt xside) XOR (l/nx mod (yside*2) ge yside)`

>

> Well, uh, it's not obvious to me. :-(

Well, `l mod (xside*2)` produces a horizontal ramp from 0... `xside*2`,
repeating over and over, the same for each row (as long as the array
width is an even multiple of `xside`). Comparing this ramp to `xside`,
ala `(ramp lt xside)` produces alternating vertical bars of width `xside`.
Similarly for `l/nx` (the row number), producing alternating horizontal
bars of height `yside`. Put them on top of each other and you have a
lovely quilt pattern. But we don't want a quilt, we want a

checkerboard, i.e. we want only those spots where the two bar patterns don't overlap, hence the exclusive or (XOR).

In case that wasn't clear, here it is dfanning style ;)

```
theRamp = LINDGEN( theNumberOfXPixels , theNumberOfYPixels )
theCheckerBoardHorizontalPeriod = 2 * theCheckWidth
theHorizontalRamp = theRamp MOD theCheckerBoardHorizontalPeriod
theHorizontalBands = theHorizontalRamp LT theCheckWidth

theRowNumbers = theRamp/theNumberOfXPixels
theCheckerBoardVerticalPeriod = 2 * theCheckHeight
theVerticalRamp = theRowNumbers MOD theCheckerBoardVerticalPeriod
theVerticalBands = theVerticalRamp GE theCheckHeight

theCheckerBoard = theHorizontalBands XOR theVerticalBands
```

JD

Subject: Re: making a checkerboard array?
Posted by [James Kuyper](#) on Wed, 26 Sep 2007 17:29:33 GMT
[View Forum Message](#) <> [Reply to Message](#)

Allan Whiteford wrote:

> Bruce Bowler wrote:

>>> http://www.nerdtests.com/ft_nq.php ;-)

>>

>>

>> Arghhhhhh 91. I didn't think I was *that* nerdy :-)

>>

>

> The test is scoring people too highly... I don't deserve the 94 I got.

> Maybe a few years ago, but not now... surely not now!?!)

I got a 97. That seems a bit high, but I'm not inclined to dispute it.

Subject: Re: making a checkerboard array?
Posted by [David Fanning](#) on Wed, 26 Sep 2007 17:39:12 GMT
[View Forum Message](#) <> [Reply to Message](#)

JD Smith writes:

> In case that wasn't clear, here it is dfanning style ;)

>

> theRamp = LINDGEN(theNumberOfXPixels , theNumberOfYPixels)

```
> theCheckerBoardHorizontalPeriod = 2 * theCheckWidth
> theHorizontalRamp = theRamp MOD theCheckerBoardHorizontalPeriod
> theHorizontalBands = theHorizontalRamp LT theCheckWidth
>
> theRowNumbers = theRamp/theNumberofXPixels
> theCheckerBoardVerticalPeriod = 2 * theCheckHeight
> theVerticalRamp = theRowNumbers MOD theCheckerBoardVerticalPeriod
> theVerticalBands = theVerticalRamp GE theCheckHeight
>
> theCheckerBoard = theHorizontalBands XOR theVerticalBands
```

Oh, hell. I think I'm going back to all lowercase. :-(

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.dfanning.com/>

Sepore ma de ni thui. ("Perhaps thou speakest truth.")

Subject: Re: making a checkerboard array?

Posted by edward.s.meinel@aero on Thu, 27 Sep 2007 14:54:43 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Sep 26, 12:44 pm, "R.G. Stockwell" <noem...@please.com> wrote:

```
> "Bruce Bowler" <bbow...@bigelow.org> wrote in message
>
> news:pan.2007.09.26.15.19.33@bigelow.org...
>
>>> http://www.nerdtests.com/ft\_nq.php;-\)
>
>> Arghhhhhh 91. I didn't think I was *that* nerdy :-)
>
> OHOH!
>
> {takes test results, prints them to printer, and burns them! Pleasantly
> dreams of
> scoring only 91}
```

99! No, that doesn't mean 'factorial'... Oooo, Agent 99, yowza!

Bruce, why are you so surprised? You're posting on comp.lang.idl-pvwave, for crying out loud!

Ed "supreme nerd god" Meinel
